* generate the additional variable  --------------------------------------------

gen coffin = 0
replace coffin = 1 if coffins == 1

* label the additional variable ------------------------------------------------

label variable coffin              "Single Treatment"

* run the regressions models and save them to memory ---------------------------

eststo: reg akp_nov coffin akp_jun recruitment_pool, cluster(province)

eststo: reg akp_nov coffin akp_jun recruitment_pool                             ///
treatment_nonterror casualty_district kurdish_district akp_district             ///
higher_education min_margin turnout_jun, cluster(province)

eststo: reg akp_nov coffin akp_jun recruitment_pool                             ///
if treatment_multi == 0, cluster(province)

eststo: reg akp_nov coffin akp_jun recruitment_pool                             ///
treatment_nonterror casualty_district kurdish_district akp_district             ///
higher_education min_margin turnout_jun if treatment_multi == 0,                ///
cluster(province)

* print the results ------------------------------------------------------------

esttab,                                                                         ///
replace se b(3) se(3) stats(N vce r2) label nodepvars nonotes nomtitles         ///
mgroups("All Districts Included" "Excluding the Multiply-Treated",              ///
pattern(1 0 1 0) span)                                                          ///
substitute(vce "Clusters" cluster "81")                                         ///
title("Table S6: Regression models with single treatment")

* save the results -------------------------------------------------------------

esttab using "./tables/table_s6.tex",                                           ///
replace se b(3) se(3) stats(N vce r2) label nodepvars nonotes nomtitles         ///
mgroups("All Districts Included" "Excluding the Multiply-Treated",              ///
pattern(1 0 1 0) prefix(\multicolumn{@span}{c}{) suffix(}) span                 ///
erepeat(\cmidrule(lr){@span}))                                                  ///
substitute(vce "Clusters" cluster "81")                                         ///
title("Regression models with single treatment")

* remove the additional variable -----------------------------------------------

drop coffin

* remove the models from memory ------------------------------------------------

eststo clear